
pkgname=expect
pkgver=5.45.4
pkgrel=1
pkgdesc="A tool for automating interactive applications"
arch=('x86_64')
url="https://www.nist.gov/el/msid/expect.cfm"
license=('custom')
depends=('tcl')
source=("https://downloads.sourceforge.net/project/expect/Expect/${pkgver}/expect${pkgver}.tar.gz")
sha256sums=('49a7da83b0bdd9f46d04a04deec19c7767bb9a323e40c4781f89caf760b92c34')

build() {
  cd expect${pkgver}

  ./configure --prefix=/usr \
    --sysconfdir=/etc \
    --localstatedir=/var \
    --mandir=/usr/share/man \
    --with-tcl=/usr/lib \
    --with-tclinclude=/usr/include

  make
}

package() {
  cd expect${pkgver}
  make DESTDIR=${pkgdir} install

  install -D -m644 README ${pkgdir}/usr/share/licenses/${pkgname}/README
}
